
GeoSPARQL

• Ontology

• query language

• OGC Standard - https://www.ogc.org/standards/geosparql

• Support in GraphDB

https://graphdb.ontotext.com/documentation/10.4/geosparql-support.html

Setup

• To enable GeoSPARQL plugin, run

PREFIX geosparql: <http://www.ontotext.com/plugins/geosparql#>

INSERT DATA {

 [] geosparql:enabled "true" .

}



Quick overview

• Download data 

from https://graphdb.ontotext.com

/documentation/10.4/_downloads/

d10860b42c7c39e3d05ef397f5756

ca1/geosparql-example.rdf and 

import them into GraphDB into the 

named graph

Quick overview – functions 

examples

Following examples are taken 

from https://graphdb.ontotext.com/documentation/standar

d/geosparql-support.html#geosparql-examples

Example 1 - Find all features that 
feature my:A contains, 
where spatial calculations are 
based on my:hasExactGeometry.



1 – solution using a function
PREFIX my: <http://example.org/ApplicationSchema#>

PREFIX geo: <http://www.opengis.net/ont/geosparql#>

PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?f

WHERE {

 my:A my:hasExactGeometry ?aGeom .

 ?aGeom geo:asWKT ?aWKT .

 ?f my:hasExactGeometry ?fGeom .

 ?fGeom geo:asWKT ?fWKT .

 FILTER (geof:sfContains(?aWKT, ?fWKT) && !sameTerm(?aGeom, ?fGeom))

}



1 – solution using a predicate

PREFIX my: <http://example.org/ApplicationSchema#>

PREFIX geo: <http://www.opengis.net/ont/geosparql#>

PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?f

WHERE {

 my:A my:hasExactGeometry ?aGeom .

 ?f my:hasExactGeometry ?fGeom .

 ?aGeom geo:sfContains ?fGeom .

 FILTER (!sameTerm(?aGeom, ?fGeom))

}



Example 2 - Find all features that 

touch the union 

of feature my:A and feature my:D, 

where computations are based 

on my:hasExactGeometry.



2 – solution using a function
PREFIX my: <http://example.org/ApplicationSchema#>

PREFIX geo: <http://www.opengis.net/ont/geosparql#>

PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?f

WHERE {

 ?f my:hasExactGeometry ?fGeom .

 ?fGeom geo:asWKT ?fWKT .

 my:A my:hasExactGeometry ?aGeom .

 ?aGeom geo:asWKT ?aWKT .

 my:D my:hasExactGeometry ?dGeom .

 ?dGeom geo:asWKT ?dWKT .

 FILTER (geof:sfTouches(?fWKT, geof:union(?aWKT, ?dWKT)))

}



2 – solution using a predicate
PREFIX my: <http://example.org/ApplicationSchema#>

PREFIX geo: <http://www.opengis.net/ont/geosparql#>

PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?f

WHERE {

 ?f my:hasExactGeometry ?fGeom .

 ?fGeom geo:asWKT ?fWKT .

 my:A my:hasExactGeometry ?aGeom .

 ?aGeom geo:asWKT ?aWKT .

 my:D my:hasExactGeometry ?dGeom .

 ?dGeom geo:asWKT ?dWKT .

 BIND(geof:union(?aWKT, ?dWKT) AS ?union) .

 ?fGeom geo:sfTouches ?union

}



Using GeoSPARQL 

with real world data



Using 

geometry in 

query

Download data for the lecture 
from https://cw.fel.cvut.cz/wiki/_media/courses/b4m
36osw/tut7.zip

Import zchu.ttl and zchu-schema.ttl into GraphDB



Topology errors

• To ignore topology errors while importing data, run

PREFIX geosparql: <http://www.ontotext.com/plugins/geosparql#>

INSERT DATA {

 [] geosparql:ignoreErrors "true" .

}



Find out if there is any 
protected area(s) on 
specific coordinates 
and return its name 
and category

14.7038 48.6663



PREFIX geo: <http://www.opengis.net/ont/geosparql#>

PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

prefix zchu: <http://osw.felk.cvut.cz/medmicha/zvláště-chráněná-území/>

prefix zchu-s: <http://osw.felk.cvut.cz/medmicha/ontologies/zvláště-chráněná-území/>

SELECT ?nazev ?typNazev

WHERE {

 ?f zchu-s:má-polohu ?fGeom ;

 zchu-s:má-název ?nazev ;

 a ?typ .

 ?typ skos:prefLabel ?typNazev .

 ?fGeom geo:asWKT ?fWKT .

 FILTER (geof:sfWithin('''<http://www.opengis.net/def/crs/OGC/1.3/CRS84> POINT (14.7038 

48.6663)'''^^geo:wktLiteral, ?fWKT))

}



Which especially protected areas lies on 

the route in file route.wkt? Divide it into 

segments per each protected area.



Steps

• Save geometry into repository (use snippet 

or INSERT keyword) - only for better 

visualization of query

• Find protected areas crossing with the 

route

• Return all points common for route and 

every single protected area (intersection)



SELECT ?zchu ?label

WHERE {

 my:route geo:hasGeometry ?myGeom .

 ?myGeom geo:asWKT ?wkt .

 ?zchu zchu-s:má-polohu ?geom ;

 zchu-s:má-název ?label .

 ?geom geo:asWKT ?zWKT .

 FILTER (geof:sfCrosses(?wkt, ?zWKT))

}
My:route is artificial feature with saved 
geometry form route.wkt file

Returns IDs and 
labels of protected 
areas crossed by 
route



SELECT ?zchu ?název ?intersection

WHERE {

 my:route geo:hasGeometry ?myGeom .

 ?myGeom geo:asWKT ?wkt .

 ?zchu zchu-s:má-polohu ?poloha ;

 zchu-s:má-název ?název .

 ?poloha geo:asWKT ?zWKT .

 FILTER (geof:sfCrosses(?wkt, ?zWKT)) .

 BIND(geof:intersection(?zWKT, ?wkt) AS ?intersection) .

}

Returns geometries 
of both protected 
area and route



For visualization try WKT playground

• https://clydedacruz.github.io/openstreetmap-wkt-playground/

• Some lines are not properly visualized – WHY?

